■ Flächenapproximationen

In[27]:=
```
Unprotect[Links]
```

Out[27]= {Links}

In[28]:=
```
Links[f_, {x_, a_, b_}] := (b - a)/n Sum[f /. x → a + k/n (b - a), {k, 0, n - 1}]
```

In[29]:=
```
LinksIntegral[f_, {x_, a_, b_}] := Limit[Links[f, {x, a, b}], n → ∞]
```

In[30]:=
```
Rechts[f_, {x_, a_, b_}] := (b - a)/n Sum[f /. x → a + k/n (b - a), {k, 1, n}]
```

In[31]:=
```
RechtsIntegral[f_, {x_, a_, b_}] := Limit[Rechts[f, {x, a, b}], n → ∞]
```

In[32]:=
```
Links[Exp[x], {x, 0, 1}]
```

Out[32]= $\frac{e-1}{\left(e^{\frac{1}{n}}-1\right)n}$

In[33]:=
```
LinksIntegral[Exp[x], {x, 0, 1}]
```

Out[33]= $e-1$

In[34]:=
```
RechtsIntegral[Exp[x], {x, 0, 1}]
```

Out[34]= $e-1$

In[35]:=
```
LinksIntegral[x^2, {x, a, b}]
```

Out[35]= $\frac{1}{3}\left(b^3-a^3\right)$

In[36]:=
```
LinksIntegral[x^3, {x, a, b}]
```

Out[36]= $\frac{1}{4}\left(b^4-a^4\right)$

In[37]:=
```
LinksIntegral[x^4, {x, a, b}]
```

Out[37]= $\frac{1}{5}\left(b^5-a^5\right)$

In[38]:=
```
RechtsGeometrisch[f_, {x_, a_, b_}] := Sum[f /. x → a * (b/a)^(k/n), {k, 1, n}] (a (b/a)^(k/n) - a (b/a)^((k-1)/n))
```

In[39]:=
```
RechtsGeometrischIntegral[f_, {x_, a_, b_}] :=
 Limit[RechtsGeometrisch[f, {x, a, b}], n → ∞]
```

In[40]:=
```
RechtsGeometrisch[1/x, {x, a, b}]
```

Out[40]= $n\left(\frac{b}{a}\right)^{-1/n}\left(\left(\frac{b}{a}\right)^{\frac{1}{n}}-1\right)$

In[41]:=
```
RechtsGeometrischIntegral[1/x, {x, a, b}]
```

Out[41]= $\log\left(\frac{b}{a}\right)$

In[42]:=
```
RechtsGeometrischIntegral[x^m, {x, a, b}]
```

Out[42]= $\frac{b^{m+1}-a^{m+1}}{m+1}$

■ Mittelwertsatz der Integralrechnung

In[43]:=
```
f[x_] := x^3 - x^2
```

In[44]:=
```
Plot[f[x], {x, -1, 2}, PlotRange -> All]
```

Out[44]=



In[45]:=
```
m = f[-1]
```

Out[45]= -2

In[46]:=
```
M = f[2]
```

Out[46]= 4

In[47]:=
```
mittelwert = (∫_-1^2 f[x] dx) / 3
```

Out[47]= $\frac{1}{4}$

In[48]:=
```
sol = Solve[f[x] == mittelwert, x]
```

Out[48]=
$$\left\{\left\{x \to \frac{1}{12}\left(4 + \sqrt[3]{280 - 24\sqrt{129}} + 2\sqrt[3]{35 + 3\sqrt{129}}\right)\right\},\right.$$
$$\left\{x \to \frac{1}{3} - \frac{1}{24}\left(1 + i\sqrt{3}\right)\sqrt[3]{280 - 24\sqrt{129}} - \frac{1}{12}\left(1 - i\sqrt{3}\right)\sqrt[3]{35 + 3\sqrt{129}}\right\},$$
$$\left.\left\{x \to \frac{1}{3} - \frac{1}{24}\left(1 - i\sqrt{3}\right)\sqrt[3]{280 - 24\sqrt{129}} - \frac{1}{12}\left(1 + i\sqrt{3}\right)\sqrt[3]{35 + 3\sqrt{129}}\right\}\right\}$$

In[49]:=
```
ξ = x /. sol[[1]]
```

Out[49]=
$$\frac{1}{12}\left(4 + \sqrt[3]{280 - 24\sqrt{129}} + 2\sqrt[3]{35 + 3\sqrt{129}}\right)$$

In[50]:=
```
N[ξ]
```

Out[50]= 1.17965

In[51]:=
```
Plot[{f[x], mittelwert}, {x, -1, 2}, PlotRange -> All,
 PlotStyle → {RGBColor[0, 0, 0], RGBColor[1, 0, 0]}]
```

Out[51]=



■ Hauptsatz der Differential- und Integralrechnung

In[52]:=
```
Clear[f]
```

In[53]:=
```
∫f'[x] dx
```

Out[53]= $f(x)$

In[54]:=
```
D[∫f[x] dx, x]
```

Out[54]= $f(x)$

In[55]:=
```
∫_a^b F'[t] dt
```

Out[55]= $F(b) - F(a)$

In[56]:=
```
D[∫_a^x f[t] dt, x]
```

Out[56]= $f(x)$

In[57]:=
```
D[∫_x^b f[t] dt, x]
```

Out[57]= $-f(x)$

In[58]:=
```
D[∫_g[x]^h[x] f[t] dt, x]
```

Out[58]= $f(h(x))\,h'(x) - f(g(x))\,g'(x)$

In[59]:=
```
∫u'[x] * v[x] dx
```

Out[59]= $\int v(x)\,u'(x)\,dx$

■ Beispiel 5.9

In[60]:=
```
int = ∫Sin[x]^2 dx
```

Out[60]= $\frac{x}{2} - \frac{1}{4}\sin(2x)$

In[61]:=
```
TrigExpand[int]
```

Out[61]= $\frac{x}{2} - \frac{1}{2}\sin(x)\cos(x)$

■ Beispiel 5.11

In[62]:=
```
int = ∫Cos[x]^3 Sin[x] dx
```

Out[62]= $-\frac{1}{4}\cos^4(x)$

■ Beispiel 5.12

In[63]:=
```
int = ∫Exp[Cos[x^2]] Sin[x^2] x dx
```

Out[63]= $-\frac{1}{2}e^{\cos(x^2)}$

■ Beispiel 5.13

In[64]:=
```
int = ∫Sqrt[1 - x^2] dx
```

Out[64]= $\frac{1}{2}\left(\sqrt{1-x^2}\ x + \sin^{-1}(x)\right)$

■ Beispiel 5.14

In[65]:=
```
int = ∫_-a^a Cosh[x]^2 dx
```

Out[65]= $a + \sinh(a)\cosh(a)$

■ Beispiel 5.15

In[66]:=
```
int = ∫_0^1 Sqrt[1 - x^2] dx
```

Out[66]= $\frac{\pi}{4}$

■ Beispiel 5.16

In[67]:=
```
int = ∫ Exp[x] / (2 Exp[x] + 1) dx
```

Out[67]= $\frac{1}{2}\log(2e^x + 1)$

■ Beispiel 5.17

In[68]:=
```
int = ∫ x / (x^2 + 1) dx
```

Out[68]= $\frac{1}{2}\log\left(x^2+1\right)$

■ Arcustangensintegral

In[69]:=
```
diff = D[x / (2 (n - 1) (1 + x^2)^(n-1)) + (2 n - 3) / (2 (n - 1)) ∫ 1 / (1 + x^2)^(n-1) dx, x]
```

Out[69]= $\frac{(2n-3)\left((x^2+1)^{1-n} - {}_2F_1\left(\frac{1}{2}, n-1; \frac{3}{2}; -x^2\right)\right)}{2(n-1)} + \frac{(2n-3)\,{}_2F_1\left(\frac{1}{2}, n-1; \frac{3}{2}; -x^2\right)}{2(n-1)} + \frac{(x^2+1)^{1-n}}{2(n-1)} + \frac{(1-n)x^2(x^2+1)^{-n}}{n-1}$

In[70]:=
```
Together[diff]
```

Out[70]= $\left(x^2+1\right)^{-n}$

■ Beispiel 5.19

In[71]:= `int = ∫ (4 x + 5)/(x^2 + 3) dx`

Out[71]= $2\log\left(x^2+3\right)+\frac{5\tan^{-1}\left(\frac{x}{\sqrt{3}}\right)}{\sqrt{3}}$

■ Beispiel 5.20

In[72]:= `int = ∫ 1/(x^4 - 1) dx`

Out[72]= $\frac{1}{4}\log(1-x)-\frac{1}{4}\log(x+1)-\frac{1}{2}\tan^{-1}(x)$

In[73]:= `Apart[1/(x^4 - 1), x]`

Out[73]= $-\frac{1}{2\left(x^2+1\right)}-\frac{1}{4(x+1)}+\frac{1}{4(x-1)}$

■ Beispiel 5.23

In[74]:= `int = ∫_0^1 1/Sqrt[1 - x] dx`

Out[74]= 2

■ Beispiel 5.24

In[75]:= `int = ∫_0^∞ x/(x^2 + 1)^2 dx`

Out[75]= $\frac{1}{2}$

■ Beispiel 5.27

In[76]:= `int = ∫_1^∞ Sin[x]/x^2 dx`

Out[76]= sin(1) − Ci(1)

In[77]:= `N[int]`

Out[77]= 0.504067

In[78]:= `NIntegrate[Sin[x]/x^2, {x, 1, ∞}]`

Out[78]= 0.504067

In[79]:=
```
Plot[Sin[x]/x^2, {x, 1, 30}]
```

Out[79]=